
11 Policy Gradient Estimation

The previous chapter discussed several ways to go about directly optimizing the
parameters of a policy to maximize expected utility. In many applications, it is
often useful to use the gradient of the utility with respect to the policy parameters
to guide the optimization process. This chapter discusses several approaches
to estimating this gradient from trajectory rollouts.1 A major challenge with

1 An additional resource on this
topic is M.C. Fu, “Gradient Estima-
tion,” in Simulation, S. G. Hender-
son and B. L. Nelson, eds., Elsevier,
2006, pp. 575–616.

this approach is the variance of the estimate due to the stochastic nature of the
trajectories arising from both the environment and our exploration of it. The next
chapter will discuss how to use these algorithms to estimate gradients for the
purpose of policy optimization.

11.1 Finite Difference

Finite difference methods estimate the gradient of a function from small changes in
its evaluation. Recall that the derivative of a univariate function f is

d f

dx
(x) = lim

δ→0

f (x + δ)− f (x)

δ
(11.1)

The derivative at x can be approximated by a sufficiently small step δ > 0:
d f

dx
(x) ≈ f (x + δ)− f (x)

δ
(11.2)

This approximation is illustrated in figure 11.1.

f (x)

x

Figure 11.1. The finite difference
method approximates the deriva-
tive of f (x) using an evaluation of
a point near x. The finite-difference
approximation, in red, is not a per-
fect match for the true derivative,
in blue.

The gradient of a multivariate function f with an input of length n is

∇ f (x) =

[

∂ f

∂x1
(x), . . . ,

∂ f

∂xn
(x)

]

(11.3)

Finite differences can be applied to each dimension to estimate the gradient.

In the context of policy optimization, we want to estimate the gradient of the
utility expected from following a policy parameterized by θ:

∇U(θ) =

[

∂U

∂θ1
(θ), . . . ,

∂U

∂θn
(θ)

]

(11.4)

≈
[

U(θ+ δe(1))−U(θ)

δ
, . . . ,

U(θ+ δe(n))−U(θ)

δ

]

(11.5)

where e(i) is the ith standard basis vector, consisting of zeros except for the ith
component, which is set to 1.

As discussed in section 10.1, we need to simulate policy rollouts to estimate
U(θ). We can use algorithm 11.1 to generate trajectories. From these trajectories,
we can compute their return and estimate the utility associated with the policy.
Algorithm 11.2 implements the gradient estimate in equation (11.5) by simulating
m rollouts for each component and averaging the returns.

function simulate(𝒫::MDP, s, π, d)
τ = []
for i = 1:d

a = π(s)
s′, r = 𝒫.TR(s,a)
push!(τ, (s,a,r))
s = s′

end
return τ

end

Algorithm 11.1. A method for gen-
erating a trajectory associated with
problem 𝒫 starting in state s and
executing policy π to depth d. It
creates a vector τ containing state-
action-reward tuples.

A major challenge in arriving at accurate estimates of the policy gradient is the
fact that the variance of the trajectory rewards can be quite high. One approach to
reduce the resulting variance in the gradient estimate is to have each rollout share
the same random generator seeds.2 This approach can be helpful, for example, 2 This random seed sharing is used

in the PEGASUS algorithm. A.Y.
Ng and M. Jordan, “A Policy
Search Method for Large MDPs
and POMDPs,” in Conference on
Uncertainty in Artificial Intelligence
(UAI), 2000.

in cases where one rollout happens to hit a low-probability transition early on.
Other rollouts will have the same tendency due to the shared random generator,
and their rewards will tend to be biased in the same way.

Policy representations have a significant effect on the policy gradient. Exam-
ple 11.1 demonstrates the sensitivity of the policy gradient to the policy parame-
terization. Finite differences for policy optimization can perform poorly when the
parameters differ in scale.

struct FiniteDifferenceGradient
𝒫 # problem
b # initial state distribution
d # depth
m # number of samples
δ # step size

end

function gradient(M::FiniteDifferenceGradient, π, θ)
𝒫, b, d, m, δ, γ, n = M.𝒫, M.b, M.d, M.m, M.δ, M.𝒫.γ, length(θ)
Δθ(i) = [i == k ? δ : 0.0 for k in 1:n]
R(τ) = sum(r*γ^(k-1) for (k, (s,a,r)) in enumerate(τ))
U(θ) = mean(R(simulate(𝒫, rand(b), s->π(θ, s), d)) for i in 1:m)
ΔU = [U(θ + Δθ(i)) - U(θ) for i in 1:n]
return ΔU ./ δ

end

Algorithm 11.2. A method for es-
timating a policy gradient using fi-
nite differences for a problem 𝒫, a
parameterized policy π(θ, s), and
a policy parameterization vector θ.
Utility estimates are made from m
rollouts to depth d. The step size is
given by δ.

Consider a single-state, single-step MDP with a one-dimensional continuous
action space and a reward function R(s, a) = a. In this case, larger actions
produce higher rewards.

Suppose we have a stochastic policy πθ that samples its action according
to a uniform distribution between θ1 and θ2 for θ2 > θ1. The expected value
is

U(θ) = E[a] =
∫ θ2

θ1

a
1

θ2 − θ1
da =

θ1 + θ2

2

The policy gradient is

∇U(θ) = [1/2, 1/2]

The policy could be reparameterized to draw actions from a uniform
distribution between θ′1 and 100θ′2, for 100θ′2 > θ′1. Now the expected reward
is (θ′1 + 100θ′2)/2 and the policy gradient is [1/2, 50].

The two parameterizations can represent the same policies, but they have
very different gradients. Finding a suitable perturbation scalar for the second
policy is much more difficult because the parameters vary widely in scale.

Example 11.1. An example of how
policy parameterization has a sig-
nificant impact on the policy gradi-
ent.

θ1 = θ′1 θ2 = 100θ′2

0

0.2

0.4

a

π
(a
|s
)

234 chapter 11. policy gradient estimation

11.2 Regression Gradient

Instead of estimating the gradient at θ by taking a fixed step along each coordinate
axis, as done in the previous section, we can use linear regression3 to estimate the 3 Linear regression is covered in

section 8.6.gradient from the results of random perturbations from θ. These perturbations
are stored in a matrix as follows:4 4 This general approach is some-

times referred to as simultaneous
perturbation stochastic approximation
by J. C. Spall, Introduction to Stochas-
tic Search and Optimization. Wiley,
2003. The general connection to
linear regression is provided by J.
Peters and S. Schaal, “Reinforce-
ment Learning ofMotor Skills with
Policy Gradients,” Neural Networks,
vol. 21, no. 4, pp. 682–697, 2008.

∆Θ =









(∆θ(1))⊤
...

(∆θ(m))⊤









(11.6)

More policy parameter perturbations will tend to produce better gradient esti-
mates.5

5 A recommended rule of thumb is
to use about twice as many pertur-
bations as the number of parame-
ters.

For each of these perturbations, we perform a rollout and estimate the change
in utility:6

6 This equation shows the forward
difference. Other finite-difference
formulations, such as the central
difference, can also be used.

∆U =
[

U(θ+ ∆θ(1))−U(θ), . . . , U(θ+ ∆θ(m))−U(θ)
]

(11.7)

The policy gradient estimate using linear regression is then7

7 As discussed in section 8.6, X+ de-
notes the pseudoinverse of X.

∇U(θ) ≈ ∆Θ
+∆U (11.8)

Algorithm 11.3 provides an implementation of this approach in which the per-
turbations are drawn uniformly from a hypersphere with radius δ. Example 11.2
demonstrates this approach with a simple function.

11.3 Likelihood Ratio

The likelihood ratio approach8 to gradient estimation uses an analytical form of 8 P.W. Glynn, “Likelihood Ratio
Gradient Estimation for Stochas-
tic Systems,” Communications of the
ACM, vol. 33, no. 10, pp. 75–84,
1990.

∇πθ to improve our estimate of ∇U(θ). Recall from equation (10.2) that

U(θ) =
∫

pθ(τ)R(τ)dτ (11.9)

struct RegressionGradient
𝒫 # problem
b # initial state distribution
d # depth
m # number of samples
δ # step size

end

function gradient(M::RegressionGradient, π, θ)
𝒫, b, d, m, δ, γ = M.𝒫, M.b, M.d, M.m, M.δ, M.𝒫.γ
ΔΘ = [δ.*normalize(randn(length(θ)), 2) for i = 1:m]
R(τ) = sum(r*γ^(k-1) for (k, (s,a,r)) in enumerate(τ))
U(θ) = R(simulate(𝒫, rand(b), s->π(θ,s), d))
ΔU = [U(θ + Δθ) - U(θ) for Δθ in ΔΘ]
return pinv(reduce(hcat, ΔΘ)') * ΔU

end

Algorithm 11.3. A method for es-
timating a policy gradient using
finite differences for an MDP 𝒫,
a stochastic parameterized policy
π(θ, s), and a policy parameter-
ization vector θ. Policy variation
vectors are generated by normal-
izing normally distributed sam-
ples and scaling by a perturbation
scalar δ. A total of m parameter per-
turbations are generated, and each
is evaluated in a rollout from an
initial state drawn from b to depth
d and compared to the original pol-
icy parameterization.

Hence,

∇U(θ) = ∇θ
∫

pθ(τ)R(τ)dτ (11.10)

=
∫

∇θpθ(τ)R(τ)dτ (11.11)

=
∫

pθ(τ)
∇θpθ(τ)

pθ(τ)
R(τ)dτ (11.12)

= Eτ

[∇θpθ(τ)

pθ(τ)
R(τ)

]

(11.13)

The name for this method comes from this trajectory likelihood ratio. This likeli-
hood ratio can be seen as a weight in likelihood weighted sampling (section 3.7)
over trajectory rewards.

Applying the log derivative trick,9 we have 9 The log derivative trick was intro-
duced in section 10.5. It uses the
following equality:
∇θ log pθ(τ) = ∇θpθ(τ)/pθ(τ)

∇U(θ) = Eτ [∇θ log pθ(τ)R(τ)] (11.14)

We can estimate this expectation using trajectory rollouts. For each trajectory
τ, we need to compute the product ∇θ log pθ(τ)R(τ). Recall that R(τ) is the
return associated with trajectory τ. If we have a stochastic policy,10 the gradient 10 We use πθ(a | s) to represent

the probability (either density or
mass) the policy πθ assigns to tak-
ing action a from state s.

∇θ log pθ(τ) is

∇θ log pθ(τ) =
d

∑
k=1

∇θ log πθ(a(k) | s(k)) (11.15)

236 chapter 11. policy gradient estimation

We would like to apply the regression gradient to estimate the gradient of
a simple, one-dimensional function f (x) = x2, evaluated at x0 = 2 from
m = 20 samples. To imitate the stochasticity inherent in policy evaluation, we
add noise to the function evaluations. We generate a set of disturbances ∆X,
sampled fromN (0, δ2), and evaluate f (x0 +∆x)− f (x0) for each disturbance
∆x in ∆X. We can then estimate the one-dimensional gradient (or derivative)
∆X+∆F with this code:
f(x) = x^2 + 1e-2*randn()
m = 20
δ = 1e-2
ΔX = [δ.*randn() for i = 1:m]
x0 = 2.0
ΔF = [f(x0 + Δx) - f(x0) for Δx in ΔX]
pinv(ΔX) * ΔF

The samples and linear regression are shown here. The slope of the regression
line is close to the exact solution of 4:

−2 −1 0 1 2

×10−2

−0.10

−0.05

0.00

0.05

0.10

∆x

∆
f

∆ f = 4.832× ∆x

Example 11.2. Using the regres-
sion gradient method on a one-
dimensional function.

because pθ(τ) takes the form

pθ(τ) = p(s(1))
d

∏
k=1

T(s(k+1) | s(k), a(k))πθ(a(k) | s(k)) (11.16)

where s(k) and a(k) are the kth state and action, respectively, in trajectory τ. Al-
gorithm 11.4 provides an implementation in which m trajectories are sampled to
arrive at a gradient estimate. Example 11.3 illustrates the process.

If we have a deterministic policy, the gradient requires computing:11 11 Many problems have vector-
valued actions a ∈ R

n. In this
case, ∇θπθ(s

(k)) is replaced with
a Jacobian matrix whose jth col-
umn is the gradient with respect
to the jth action component, and
the ∂

∂a(k)
log T(s(k+1) | s(k), a(k)) is

replaced with an action gradient.

∇θ log pθ(τ) = ∇θ log

[

p(s(1))
d

∏
k=1

T(s(k+1) | s(k), πθ(s
(k)))

]

(11.17)

=
d

∑
k=1

∇θπθ(s
(k))

∂

∂a(k)
log T(s(k+1) | s(k), a(k)) (11.18)

Equations (11.17) and (11.18) require knowing the transition likelihood, which is
in contrast with equation (11.15) for stochastic policies.

struct LikelihoodRatioGradient
𝒫 # problem
b # initial state distribution
d # depth
m # number of samples
∇logπ # gradient of log likelihood

end

function gradient(M::LikelihoodRatioGradient, π, θ)
𝒫, b, d, m, ∇logπ, γ = M.𝒫, M.b, M.d, M.m, M.∇logπ, M.𝒫.γ
πθ(s) = π(θ, s)
R(τ) = sum(r*γ^(k-1) for (k, (s,a,r)) in enumerate(τ))
∇U(τ) = sum(∇logπ(θ, a, s) for (s,a) in τ)*R(τ)
return mean(∇U(simulate(𝒫, rand(b), πθ, d)) for i in 1:m)

end

Algorithm 11.4. A method for esti-
mating a policy gradient of a pol-
icy π(s) for an MDP 𝒫 with initial
state distribution b using the likeli-
hood ratio trick. The gradient with
respect to the parameterization vec-
tor θ is estimated from m rollouts to
depth d using the log policy gradi-
ents ∇logπ.

11.4 Reward-to-Go

The likelihood ratio policy gradient method is unbiased but has high variance.
Example 11.4 reviews bias and variance. The variance generally increases sig-
nificantly with rollout depth due to the correlation between actions, states, and

238 chapter 11. policy gradient estimation

Consider the single-step, single-state problem from example 11.1. Suppose
we have a stochastic policy πθ that samples its action according to a Gaussian
distribution N (θ1, θ2

2), where θ2
2 is the variance.

log πθ(a | s) = log





1
√

2πθ2
2

exp

(

− (a− θ1)
2

2θ2
2

)





= − (a− θ1)
2

2θ2
2

− 1

2
log
(

2πθ2
2

)

The gradient of the log policy likelihood is

∂

∂θ1
log πθ(a | s) =

a− θ1

θ2
2

∂

∂θ2
log πθ(a | s) =

(a− θ1)
2 − θ2

2

θ3
2

Suppose we run three rollouts with θ = [0, 1], taking actions {0.5,−1, 0.7}
and receiving the same rewards (R(s, a) = a). The estimated policy gradient
is

∇U(θ) ≈ 1

m

m

∑
i=1

∇θ log pθ(τ
(i))R(τ(i))

=
1

3

([

0.5

−0.75

]

0.5 +

[

−1.0

0.0

]

(−1) +

[

0.7

−0.51

]

0.7

)

= [0.58,−0.244]

Example 11.3. Applying the likeli-
hood ratio trick to estimate a policy
gradient in a simple problem.

When estimating a quantity of interest from a collection of simulations, we
generally want to use a scheme that has both low bias and low variance. In
this chapter, we want to estimate ∇U(θ). Generally, with more simulation
samples, we can arrive at a better estimate. Some methods can lead to bias,
where—even with infinitely many samples—it does not lead to an accurate
estimate. Sometimes methods with nonzero bias may still be attractive if they
also have low variance, meaning that they require fewer samples to converge.

Here are plots of the estimates from four notional methods for estimating
∇U(θ). The true value is 17.5, as indicated by the red lines. We ran 100

simulations 100 times for eachmethod. The variance decreases as the number
of samples increases. The blue regions indicate the 5 % to 95 % and 25 % to
75 % empirical quantiles of the estimates.

10

15

20

25

hi
gh

bi

∇
U
(θ
)

high variance low variance

100 101 102
10

15

20

25

number of samples

low
bi
as

∇
U
(θ
)

100 101 102

number of samples

Example 11.4. An empirical
demonstration of bias and
variance when estimating ∇U(θ).

240 chapter 11. policy gradient estimation

rewards across time steps. The reward-to-go approach attempts to reduce the
variance in the estimate.

To derive this approach, we begin by expanding equation (11.14):

∇U(θ) = Eτ

[(

d

∑
k=1

∇θ log πθ(a(k) | s(k))

)(

d

∑
k=1

r(k)γk−1

)]

(11.19)

Let f (k) replace ∇θ log πθ(a(k) | s(k)) for convenience. We then expand as
follows:

∇U(θ) = Eτ

[(

d

∑
k=1

f (k)

)(

d

∑
k=1

r(k)γk−1

)]

(11.20)

= Eτ

[(

f (1) + f (2) + f (3) + · · ·+ f (d)
)(

r(1) + r(2)γ + r(3)γ2 + · · ·+ r(d)γd−1
)]

(11.21)

= Eτ

















f (1)r(1) + f (1)r(2)γ + f (1)r(3)γ2 + · · ·+ f (1)r(d)γd−1

+ f (2)r(1) + f (2)r(2)γ + f (2)r(3)γ2 + · · ·+ f (2)r(d)γd−1

+ f (3)r(1) + f (3)r(2)γ + f (3)r(3)γ2 + · · ·+ f (3)r(d)γd−1

...
+ f (d)r(1) + f (d)r(2)γ + f (d)r(3)γ2 + · · ·+ f (d)r(d)γd−1

















(11.22)

The first reward, r(1), is affected only by the first action. Thus, its contribution to
the policy gradient should not depend on subsequent time steps. We can remove
other such causality-violating terms as follows:12 12 The term ∑

d
ℓ=k r(ℓ)γℓ−k is often

called the reward-to-go from step k.

∇U(θ) = Eτ

















f (1)r(1) + f (1)r(2)γ + f (1)r(3)γ2 + · · ·+ f (1)r(d)γd−1

+ f (2)r(2)γ + f (2)r(3)γ2 + · · ·+ f (2)r(d)γd−1

+ f (3)r(3)γ2 + · · ·+ f (3)r(d)γd−1

...
+ f (d)r(d)γd−1

















(11.23)

= Eτ

[

d

∑
k=1

∇θ log πθ(a(k) | s(k))

(

d

∑
ℓ=k

r(ℓ)γℓ−1

)]

(11.24)

= Eτ

[

d

∑
k=1

∇θ log πθ(a(k) | s(k))

(

γk−1
d

∑
ℓ=k

r(ℓ)γℓ−k

)]

(11.25)

= Eτ

[

d

∑
k=1

∇θ log πθ(a(k) | s(k))γk−1r
(k)
to-go

]

(11.26)

Algorithm 11.5 provides an implementation of this.
Notice that the reward-to-go for a state-action pair (s, a) under a policy param-

eterized by θ is really an approximation of the state-action value from that state,
Qθ(s, a). The action value function, if known, can be used to obtain the policy
gradient:

∇U(θ) = Eτ

[

d

∑
k=1

∇θ log πθ(a(k) | s(k))γk−1Qθ

(

s(k), a(k)
)

]

(11.27)

struct RewardToGoGradient
𝒫 # problem
b # initial state distribution
d # depth
m # number of samples
∇logπ # gradient of log likelihood

end

function gradient(M::RewardToGoGradient, π, θ)
𝒫, b, d, m, ∇logπ, γ = M.𝒫, M.b, M.d, M.m, M.∇logπ, M.𝒫.γ
πθ(s) = π(θ, s)
R(τ, j) = sum(r*γ^(k-1) for (k,(s,a,r)) in zip(j:d, τ[j:end]))
∇U(τ) = sum(∇logπ(θ, a, s)*R(τ,j) for (j, (s,a,r)) in enumerate(τ))
return mean(∇U(simulate(𝒫, rand(b), πθ, d)) for i in 1:m)

end

Algorithm 11.5. A method that
uses reward-to-go for estimating
a policy gradient of a policy π(s)
for an MDP 𝒫 with initial state dis-
tribution b. The gradient with re-
spect to the parameterization vec-
tor θ is estimated from m rollouts to
depth d using the log policy gradi-
ent ∇logπ.

11.5 Baseline Subtraction

We can further build on the approach presented in the previous section by sub-
tracting a baseline value from the reward-to-go13 to reduce the variance of the 13 We could also subtract a baseline

from a state-action value.gradient estimate. This subtraction does not bias the gradient.
We now subtract a baseline rbase(s(k)):

∇U(θ) = Eτ

[

d

∑
k=1

∇θ log πθ(a(k) | s(k))γk−1
(

r
(k)
to-go − rbase(s

(k))
)

]

(11.28)

To show that baseline subtraction does not bias the gradient, we first expand:

∇U(θ) = Eτ

[

d

∑
k=1

∇θ log πθ(a(k) | s(k))γk−1r
(k)
to-go −

d

∑
k=1

∇θ log πθ(a(k) | s(k))γk−1rbase(s
(k))

]

(11.29)

242 chapter 11. policy gradient estimation

The linearity of expectation states that E[a + b] = E[a] + E[b], so it is sufficient
to prove that equation (11.29) is equivalent to equation (11.26), if for each step k,
the expected associated baseline term is 0:

Eτ

[

∇θ log πθ(a(k) | s(k))γk−1rbase(s
(k))
]

= 0 (11.30)
We begin by converting the expectation into nested expectations, as illustrated

in figure 11.2:

Eτ

[

∇θ log πθ(a(k) | s(k))γk−1rbase(s
(k))
]

= Eτ1:k

[

Eτk+1:d

[

∇θ log πθ(a(k) | s(k))γk−1rbase(s
(k))
]]

(11.31)

Eτ [ f (τ)] Eτ1:k

[

Eτk+1:d
[ f (τ)]

]

=

Figure 11.2. The expectation of
a function of trajectories sampled
from a policy can be viewed as
an expectation over a nested ex-
pectation of subtrajectories. For a
mathematical derivation, see exer-
cise 11.4.

We continue with our derivation, using the same log derivative trick from
section 11.3:

Eτ1:k

[

Eτk+1:d

[

∇θ log πθ(a(k) | s(k))γk−1rbase(s
(k))
]]

= Eτ1:k

[

γk−1rbase(s
(k)) Eτk+1:d

[

∇θ log πθ(a(k) | s(k))
]]

(11.32)

= Eτ1:k

[

γk−1rbase(s
(k)) Ea(k)

[

∇θ log πθ(a(k) | s(k))
]]

(11.33)

= Eτ1:k

[

γk−1rbase(s
(k))

∫

∇θ log πθ(a(k) | s(k))πθ(a(k) | s(k))da(k)
]

(11.34)

= Eτ1:k

[

γk−1rbase(s
(k))

∫ ∇θπθ(a(k) | s(k))

πθ(a(k) | s(k))
πθ(a(k) | s(k))da(k)

]

(11.35)

= Eτ1:k

[

γk−1rbase(s
(k))∇θ

∫

πθ(a(k) | s(k))da(k)
]

(11.36)

= Eτ1:k

[

γk−1rbase(s
(k))∇θ1

]

(11.37)

= Eτ1:k

[

γk−1rbase(s
(k)) 0

]

(11.38)

11.5. baseline subtraction 243

Therefore, subtracting a term rbase(s(k)) does not bias the estimate. This derivation
assumed continuous state and action spaces. The same result applies to discrete
spaces.

We can choose a different rbase(s) for every component of the gradient, and
we will select them to minimize the variance. For simplicity, we will drop the de-
pendence on s and treat each baseline component as constant.14 For compactness

14 Some methods approximate a
state-dependent baseline using
rbase(s(k)) = φ(s(k))⊤w. Selecting
appropriate baseline functions
tends to be difficult. J. Peters
and S. Schaal, “Reinforcement
Learning of Motor Skills with
Policy Gradients,” Neural Networks,
vol. 21, no. 4, pp. 682–697, 2008.

in writing the equations in our derivation, we define

ℓi(a, s, k) = γk−1 ∂

∂θi
log πθ(a | s) (11.39)

The variance of the ith component of our gradient estimate in equation (11.28) is

E
a,s,rto-go,k

[

(

ℓi(a, s, k)
(

rto-go − rbase,i

))2
]

− E
a,s,rto-go,k

[

ℓi(a, s, k)
(

rto-go − rbase,i

)]2 (11.40)

where the expectation is over the (a, s, rto-go) tuples in our trajectory samples, and
k is each tuple’s depth.

We have just shown that the second term is zero. Hence, we can focus on
choosing rbase,i to minimize the first term by taking the derivative with respect to
the baseline and setting it to zero:

∂

∂rbase,i
E

a,s,rto-go,k

[

(

ℓi(a, s, k)
(

rto-go − rbase,i

))2
]

=
∂

∂rbase,i

(

E
a,s,rto-go,k

[

ℓi(a, s, k)2r2
to-go

]

− 2 E
a,s,rto-go,k

[

ℓi(a, s, k)2rto-gorbase,i

]

+ r2
base,i E

a,s,k

[

ℓi(a, s, k)2
]

)

(11.41)
= −2 E

a,s,rto-go,k

[

ℓi(a, s, k)2rto-go
]

+ 2rbase,i E
a,s,k

[

ℓi(a, s, k)2
]

= 0 (11.42)

Solving for rbase,i yields the baseline component that minimizes the variance:

rbase,i =
Ea,s,rto-go,k

[

ℓi(a, s, k)2rto-go
]

Ea,s,k

[

ℓi(a, s, k)2
] (11.43)

244 chapter 11. policy gradient estimation

struct BaselineSubtractionGradient
𝒫 # problem
b # initial state distribution
d # depth
m # number of samples
∇logπ # gradient of log likelihood

end

function gradient(M::BaselineSubtractionGradient, π, θ)
𝒫, b, d, m, ∇logπ, γ = M.𝒫, M.b, M.d, M.m, M.∇logπ, M.𝒫.γ
πθ(s) = π(θ, s)
ℓ(a, s, k) = ∇logπ(θ, a, s)*γ^(k-1)
R(τ, k) = sum(r*γ^(j-1) for (j,(s,a,r)) in enumerate(τ[k:end]))
numer(τ) = sum(ℓ(a,s,k).^2*R(τ,k) for (k,(s,a,r)) in enumerate(τ))
denom(τ) = sum(ℓ(a,s,k).^2 for (k,(s,a)) in enumerate(τ))
base(τ) = numer(τ) ./ denom(τ)
trajs = [simulate(𝒫, rand(b), πθ, d) for i in 1:m]
rbase = mean(base(τ) for τ in trajs)
∇U(τ) = sum(ℓ(a,s,k).*(R(τ,k).-rbase) for (k,(s,a,r)) in enumerate(τ))
return mean(∇U(τ) for τ in trajs)

end

Algorithm 11.6. Likelihood ratio
gradient estimation with reward-
to-go and baseline subtraction for
anMDP 𝒫, policy π, and initial state
distribution b. The gradient with
respect to the parameterization vec-
tor θ is estimated from m rollouts to
depth d using the log policy gradi-
ents ∇logπ.

Figure 11.3. Several policy gradient
methods used to optimize policies
for the simple regulator problem
from the same initial parameteriza-
tion. Each gradient evaluation ran
six rollouts to depth 10. The magni-
tude of the gradient was limited to
1, and step updates were applied
with step size 0.2. The optimal pol-
icy parameterization is shown in
black.

11.6. summary 245

It is common to use likelihood ratio policy gradient estimation with this base-
line subtraction (algorithm 11.6).15. Figure 11.3 compares the methods discussed 15 This combination is used in

the class of algorithms called
REINFORCE as introduced by
R. J. Williams, “Simple Statistical
Gradient-Following Algorithms
for Connectionist Reinforcement
Learning,”Machine Learning, vol. 8,
pp. 229–256, 1992.

here.
Qualitatively, when considering the gradient contribution of state-action pairs,

what we really care about is the relative value of one action over another. If all
actions in a particular state produce the same high value, there is no real signal
in the gradient, and baseline subtraction can zero that out. We want to identify
the actions that produce a higher value than others, regardless of the mean value
across actions.

An alternative to the action value is the advantage, A(s, a) = Q(s, a)−U(s).
Using the state value function in baseline subtraction produces the advantage.
The policy gradient using the advantage is unbiased and typically has much lower
variance. The gradient computation takes the following form:

∇U(θ) = Eτ

[

d

∑
k=1

∇θ log πθ(a(k) | s(k))γk−1 Aθ

(

s(k), a(k)
)

]

(11.44)

As with the state and action value functions, the advantage function is typically
unknown. Other methods, covered in chapter 13, are needed to approximate it.

11.6 Summary

• A gradient can be estimated using finite differences.

• Linear regression can also be used to provide more robust estimates of the
policy gradient.

• The likelihood ratio can be used to derive a form of the policy gradient that
does not depend on the transition model for stochastic policies.

• The variance of the policy gradient can be significantly reduced using the
reward-to-go and baseline subtraction.

246 chapter 11. policy gradient estimation

11.7 Exercises
Exercise 11.1. If we estimate the expected discounted return of a given parameterized
policy πθ defined by an n-dimensional vector of parameters θ using m rollouts, how many
total rollouts dowe need to perform to compute the policy gradient using a finite difference
approach?

Solution: In order to estimate the policy gradient using a finite difference approach, we
need to estimate the utility of the policy given the current parameter vector U(θ), as well as
all n variations of the current parameter vector U(θ+ δe(i)) for i = 1 : n. Since we estimate
each of these using m rollouts, we need to perform a total of m(n + 1) rollouts.

Exercise 11.2. Suppose we have a robotic arm with which we are able to run experiments
manipulating a wide variety of objects. We would like to use the likelihood ratio policy
gradient or one of its extensions to train a policy that is efficient at picking up and moving
these objects. Would it be more straightforward to use a deterministic or a stochastic policy,
and why?

Solution: The likelihood ratio policy gradient requires an explicit representation of the
transition likelihood when used with deterministic policies. Specifying an accurate explicit
transition model for a real-world robotic arm manipulation task would be challenging.
Computing the policy gradient for a stochastic policy does not require having an explicit
representation of the transition likelihood, making the use of a stochastic policy more
straightforward.

Exercise 11.3. Consider policy gradients of the form

∇θU(θ) = Eτ

[

d

∑
k=1

γk−1y∇θ log πθ(a(k) | s(k))

]

Which of the following values of y result in a valid policy gradient? Explain why.
(a) γ1−k ∑

∞
ℓ=1 r(ℓ)γℓ−1

(b) ∑
∞
ℓ=k r(ℓ)γℓ−k

(c)
(

∑
∞
ℓ=k r(ℓ)γℓ−k

)

− rbase(s(k))

(d) U(s(k))

(e) Q(s(k), a(k))

(f) A(s(k), a(k))

(g) r(k) + γU(s(k+1))−U(s(k))

Solution:
(a) ∑

∞
ℓ=1 r(ℓ) results in the total discounted reward, as

γk−1γ1−k
∞

∑
ℓ=1

r(ℓ)γℓ−1 =
∞

∑
ℓ=1

r(ℓ)γℓ−1

and produces a valid policy gradient, as given in equation (11.19).
(b) ∑

∞
ℓ=k r(ℓ)γℓ−k is the reward-to-go and produces a valid policy gradient, as given in

equation (11.26).
(c)

(

∑
∞
ℓ=k r(ℓ)

)

− rbase(s(k)) is the baseline subtracted reward-to-go and produces a
valid policy gradient, as given in equation (11.28).

(d) U(s(k)) is the state value function and does not produce a valid policy gradient.
(e) Q(s(k), a(k)) is the state-action value function and produces a valid policy gradient,

as given in equation (11.27).
(f) A(s(k), a(k)) is the advantage function and produces a valid policy gradient, as given

in equation (11.44).
(g) r(k) + γU(s(k+1)) − U(s(k)) is the temporal difference residual (to be discussed

further in chapter 13) and produces a valid policy gradient because it is an unbiased
approximation of the advantage function.

Exercise 11.4. Show that Eτ∼π [ f (τ)] = Eτ1:k∼π [Eτk:d∼π [ f (τ)]] for step k.
Solution: The nested expectations can be proven by writing the expectation in integral form
and then converting back:

Eτ∼π [ f (τ)] =

=
∫

p(τ) f (τ)dτ

=
∫

(

p(s(1))
d

∏
k=1

p(s(k+1) | s(k), a(k))π(a(k) | s(k))

)

f (τ)dτ

=
∫ ∫ ∫ ∫

· · ·
∫

(

p(s(1))
d

∏
k=1

p(s(k+1) | s(k), a(k))π(a(k) | s(k))

)

f (τ)ds(d) · · ·da(2) ds(2) da(1) ds(1)

= E
τ1:k∼π





∫ ∫ ∫ ∫

· · ·
∫





d

∏
q=k

p(s(q+1) | s(q), a(q))π(a(q) | s(q))



 f (τ)ds(d) · · ·da(k+1) ds(k+1) da(k) ds(k)





= Eτ1:k∼π [Eτk:d∼π [ f (τ)]]

Exercise 11.5. Our implementation of the regression gradient (algorithm 11.3) fits a
linear mapping from perturbations to the difference in returns, U(θ+ ∆θ(i))−U(θ). We
evaluate U(θ+ ∆θ(i)) and U(θ) for each of the m perturbations, thus reevaluating U(θ)

a total of m times. How might we reallocate the samples in a more effective manner?

248 chapter 11. policy gradient estimation

Solution: One approach is to evaluate U(θ) once and use the same value for each perturba-
tion, thereby conducting only m + 1 evaluations. Having an accurate estimate of U(θ) is
particularly important for an accurate regression gradient estimate. An alternative is to
still compute U(θ) once, but use m rollouts, thus preserving the total number of rollouts
per iteration. This approach uses the same amount of computation as algorithm 11.3, but
it can produce a more reliable gradient estimate.

